

ANSYS 11.0 New Feature

接触要素による剥離

2008/03/31

サイバネットシステム株式会社



CYBERNET **ANSYS Classicにおける剥離挙動**

ANSYSにおいて剥離を表現する方法として以下の2つの方法がある。


1.接触要素による剥離 - **ANSYS11.0 New Feature** -
CONTA171～CONTA177を使用

2.インターフェース要素による剥離 - **ANSYS10.0 New Feature** -
INTER202～INTER205を使用

“接触要素”と“インターフェース要素”の剥離機能の特徴

1. 接触要素による剥離挙動


- ・二直線材料挙動
- ・不連続メッシュ間の剥離が可能
- ・材料定数の算出が容易
- ・既存の接触要素に材料を追加することにより設定が可能
- ・既存の接触要素の結果情報(接触圧力、ギャップ、滑り量など)による結果評価が可能



※本資料では、接触要素による剥離挙動について紹介


2. インターフェース要素による剥離挙動

- ・EXPを用いた材料挙動
- ・剥離面においてメッシュの整合性が必要




表現できる剥離モード

- ・剥離モードⅠ 法線方向の剥離
- ・剥離モードⅡ 接線方向の剥離
- ・剥離モードⅢ 法線方向+接線方向の剥離



剥離モードⅠ




剥離モードⅡ

剥離の材料モデルでは、**分離距離 (CBDD)** で指定する方法と
臨界破壊エネルギー (CBDE) で指定する2種類がある。

表面力と分離距離を持つ2直線材料挙動 (CBDD)

TB,CZM,材料番号,1,1,CBDD
TBDATA,1,c1,c2,c3,c4,c5,c6

- c1 — σ_{\max} : 法線方向の接触応力の最大値
- c2 — U_n^c : 剥離終了時の接触ギャップ
- c3 — τ_{\max} : 接線方向の接触応力と等価の最大値
- c4 — U_t^c : 剥離終了時の接線方向の滑り
- c5 — η : 擬似減衰係数
- c6 — β : 法線方向の圧縮性接触応力に基づく
接線方向の滑りフラグ




- ・法線方向のみの剥離では、最低限 c1,c2 とリアルコンスタントFKNの設定が必要。
- ・c5,c6はオプション。
- ・接線方向の剥離では、最低限 c3,c4 とリアルコンスタントFKTの設定が必要。
- ・ギャップ量が、 U_n^c を超えると完全に剥離して接触面に応力は生じない。

表面力と臨界破壊エネルギーを持つ2直線材料挙動 (CBDE)

TB,CZM,材料番号,1,1,CBDE
TBDATA,1,c1,c2,c3,c4,c5,c6


- c1 — σ_{\max} : 法線方向の接触応力の最大値
- c2 — G_n^c : 法線方向分離のための臨界破壊エネルギー
- c3 — τ_{\max} : 接線方向の接触応力と等価の最大値
- c4 — G_t^c : 接線方向滑りのための臨界破壊エネルギー
- c5 — η : 擬似減衰係数
- c6 — β : 法線方向の圧縮性接触応力に基づく
接線方向の滑りフラグ



- ・CBDDとの違いは、剥離量(GAP)ではなくエネルギーで判定する点のみ。
- ・ G_n^c を臨界破壊エネルギーとして、**エネルギーが G_n^c を超えると完全に剥離して**接触面に応力は生じない。
 $G_n^c = 0.5 \cdot \sigma_{\max} \cdot U_n^c$

作成手順

- ①.剥離面を節点共有しない状態にてモデルを作成
 - ②.接触要素の定義
 - ③.粘着領域材料モデル TB,CZMの定義
 - ④.接触要素を生成(コンタクト面、ターゲット面)
TBコマンドで粘着領域材料を設定した材料番号をアクティブにすること
- ※手順③において粘着領域の材料を定義する以外は、一般的な接触要素の設定手順と同じ。



利用可能な接触要素タイプ

- ・面一面 (CONTA171~CONTA174)
 - ・点一面 (CONTA175)
 - ・線ー線 (CONTA176)
 - ・線ー面 (CONTA177)
- ※点一点以外の接触要素は全て利用可能。

利用可能なキーオプションの設定

- 1.接触アルゴリズム
 - ・拡大ラグランジェ法 KEYOPT(2)=0
 - ・ペナルティ法 KEYOPT(2)=1
- 2.接触面の挙動
 - ・分離なし(常時) KEYOPT(12)=4
 - ・固着(常時) KEYOPT(12)=5
 - ・固着(初期接触) KEYOPT(12)=6

出力される結果項目


- ・ 接触要素における出力データは利用可能
(接触圧、ギャップ、滑り量等)
- ・ 法線方向の粘着力は、接触圧力(CONT,PRES)で、法線方向の剥離量は、ギャップ量(CONT,GAP)で確認できる。
- ・ 接線方向の粘着力は、接触応力(CONT,SFRIC)で、接線方向の剥離量は、滑り量(CONT,SLIDE)で確認できる。

以下の結果項目は、“接触要素による剥離”でのみ出力される結果項目であり要素テーブルにより取得可能。

- ・ 時刻歴剥離(DTSTART)
- ・ 剥離パラメータ(dn,dt,dm)
- ・ 臨界破壊エネルギー(DENERI,DENERIⅡ)

剥離の例題

2つのエリア間に粘着領域材料を設定し、強制変位により粘着面を剥離させる。




----- モデル概要 -----

- ・ 要素タイプ Plane182
- ・ 2次元モデル(平面ひずみ)
- ・ TB,CZM 表面力と臨界破壊エネルギーを持つ2直線材料挙動(CBDE)を設定

----- 荷重条件 -----

- ・ 材料1の左端部を完全固定
- ・ 材料2の左端部についてUX方向を固定。UY方向に強制変位(0.01)。



変形後

! インプット(ex1.inp)

! ----- 作成手順 -----

!①剥離面を節点共有しない状態にてモデルを作成

!②接触要素を定義

!③粘着領域材料モデル TB,CZMの定義

!④接触要素の生成(コンタクト面、ターゲット面)

!

! ----- Parameter Set -----

c1=800.0 !法線方向の接触応力の最大値

c2=0.4 !剥離終了時の接触ギャップ

c5=1e-5 !擬似減衰係数

kslope=2e7 !接触剛性FKN

pinb=c2*2/c1+1e-5 !ピンボール半径を指定

! -----

!①剥離面を節点共有しない状態にてモデルを作成

/prep7

et,1,182

keyopt,1,1,2 !強化ひずみ定式化

keyopt,1,3,2 !平面ひずみ

mp,ex,1,6e6 !材料1 ヤング率

mp,prxy,1,0.49 !材料1 ポアソン比

mp,ex,2,7e10 !材料2 ヤング率

mp,prxy,2,0.33 !材料2 ポアソン比

rect,,0.1,0,0.0025

aatt,1,1,1

asel,none

rect,,0.1,0.0025,0.005

aatt,2,1,1

allsel

esize,0.001

amesh,all

!要素分割を実行

!②接触要素を定義

et,2,169

!接触要素の定義(ターゲット要素)

et,3,171

!接触要素の定義(コンタクト要素)

keyopt,3,12,5

!接触面の挙動(常に固着)

r,3,,,kslope,,,pinb

!リアルコンスタントの定義

!③粘着領域材料モデル TB,CZMの定義

tb,czm,3,1,1,cbde

!粘着材料の定義

tbdata,1,c1,c2,,,c5

!④接触要素の生成(コンタクト面、ターゲット面)

type,2

mat,3

real,3

esel,s,mat,,1

nsle,s

nsel,r,loc,y,0.0025

esurf

!接触要素の生成(ターゲット面)

type,3

mat,3

real,3

esel,s,mat,,2

nsle,s

nsel,r,loc,y,0.0025

esurf !接触要素の生成(コンタクト面)

allsel

finish

/solu

antype,static

nlgeom,on

!大変形の設定

nsubst,200,500,50

!サブステップの設定

outres,all,all

esel,s,mat,,1

nsle,s

nsel,r,loc,x,0

d,all,all,0 !材料1の左端部を固定

esel,s,mat,,2

nsle,s

nsel,r,loc,x,0

d,all,ux,0 !材料2の左端部のUX方向を固定

d,all,uy,0.01

!材料2の左端部のUY方向に強制変位

allsel

solve

!解析を実行

finish

/post1


set,last

pldisp,2

!変形形状を表示

・ 接触剛性について

粘着領域材料における接触剛性(FKN,FKT)は、slopeを意味するためデフォルトの自動算出ではなく、**絶対値**で指定すること。



・ TB,CZMの材料入力について

TB,CZMの物性値c1からc6については必ずしも全て入力する必要はない。
例えば、剥離モードⅠのみの挙動を表す場合は、c1,c2,c5のみを入力。

- : 剥離モードⅠ (法線)に関連 — c1,c2,c5
- : 剥離モードⅡ (接線)に関連 — c3,c4,c5
- : 剥離モードⅢ (混合)に関連 — c1~c6

・ 擬似減衰係数 η について(c5)

擬似減衰係数 η は、ニュートンラプソンの計算における収束安定性のために使用される。特にギャップ量が、 U_n や U_n^c では、接触剛性の変化により解析が不安定になりやすいため、擬似減衰係数を定義することが推奨される。


擬似減衰係数 η は、最小時間増分量 Δt の 0.1～0.01 倍程度が目安となる。

・ 剥離パラメータ d について

粘着領域材料では、ギャップ量が \overline{U}_n を超えた時点から剥離が始まると考える。一度、剥離が生じると、粘着領域における粘着力は低下してギャップ量が U_n^c に達すると粘着力はゼロとなる。

また、一度剥離が生じると、途中で荷重を除荷して接触面間が再度くっついても粘着力は低下したままである。

それを表現するのが剥離パラメータ d である。



剥離パラメータ d は、下記式で組み込まれる。

$$P = K_n U_n (1 - d_n)$$

$$d_n = \left(\frac{U_n - \bar{U}_n}{U_n} \right) \left(\frac{U_n^c}{U_n^c - \bar{U}_n} \right)$$

$$\frac{U_n}{\bar{U}_n} \leq 1 \quad d_n = 0$$

$$\frac{U_n}{\bar{U}_n} > 1 \quad 0 < d_n \leq 1$$

P = 法線方向の接触圧力


K_n = 法線方向の接触剛性

U_n = 接触面のギャップ



d_n = 法線方向の剥離パラメータ

\bar{U}_n = 最大接触圧力が生じるギャップ量

U_n^c = 完全に剥離するギャップ量




剥離パラメータ d の効果 $\text{Slope} = K_n(1 - d_n)$



ギャップ量が U_n^c を超えると、完全に剥離した状態になる。この時 $d_n = 1$ となるため Slope はゼロとなる。以降、接触圧力は生じない。

・ピンボール領域 について

ギャップ量がピンボール半径を超えると、 U_n^c に達していなくても、接触圧力はゼロになる。その際は接触圧力が急激に変化するため解析の発散を招く可能性が高い。



・初期ギャップ について

接触要素による剥離機能では、ギャップ量に応じた粘着力が生じる。そのため、初期状態で接触面間にギャップが存在すると、初期状態において接触面間に粘着力が生じる。また、初期で大きなギャップが存在する場合は、剥離パラメータ d の効果により意図した挙動をとらない可能性がある。

対応策の一つとして、**コンタクト要素のKEYOPT(9)=1 "形状的初期食い込み/ギャップとオフセットの両方を除外する"**により初期ギャップを除外する方法が考えられる。

・ 法線方向の圧縮力に基づく接線方向の滑りフラグ β について(c6)

接触面に法線方向の圧縮力が生じている時に、剥離モードⅡが生じるかを設定するフラグ(c6にて設定)。

$\beta = 0$ or 1 のどちらかを入力。

- $\beta = 0$: 接触面に対して法線方向の圧縮力が生じている時は接線方向に力が作用しても剥離モードⅡが生じない。
 $\beta = 1$: 接触面に対して法線方向の圧縮力が生じている時に接線方向の力が作用すると剥離モードⅡが生じる。

本オプションは、接触面に対して垂直方向の引張り力が働いている時には効きません。引張り力が働いている時は、C6の設定に関わらず接線方向に力が作用すると剥離が生じます。

・ GUIメニューにおける粘着領域材料モデルの設定

粘着領域材料モデルは、GUIメニューに対応していない。そのため、材料の設定(TB,CZM)については、コマンドで行う必要がある。

ただし、接触要素の定義や接触のオプション設定など、材料の設定(TB,CZM)以外の手順については、GUIメニューより設定が可能である。

GUIメニューにおける作成手順

- ①.剥離面を節点共有しない状態にてモデルを作成
- ②.コンタクトウィザードを用いて接触を定義
- ③.コンタクトウィザードを用いて接触のオプションを設定
 - ・接触アルゴリズムは、“ペナルティ法”、“拡大ラグランジェ法”から選択
 - ・接触面の挙動で、“常に分離なし”、“常に固着”、“初期のみ固着”を選択
 - ・垂直方向および接線方向ペナルティ剛性を定数(絶対値)で設定
- ④.粘着領域材料モデル TB,CZMの定義
※手順④のみコマンドで実施 P.5,6を参照。